

Examples of computation of parameterized differential Galois groups for some second-order linear differential equations with one differential parameter

Carlos E. Arreche
January 7, 2015

This note is a companion to the author's papers

[1] C.A., *A Galois-theoretic proof of the differential transcendence of the incomplete Gamma function*, J. Algebra **389**, 119-127 (2013);

[2] C.A., *Computation of the unipotent radical of the differential Galois group of a parameterized second-order linear differential equation*, Adv. Appl. Math. **57**, 44-59 (2014);

[3] C.A., *Computing the differential Galois group of a parameterized second-order linear differential equation*, Proceedings of ISSAC 2014, 43-50 (2014).

Together with

[4] T. Dreyfus, *Computing the Galois group of some parameterized linear differential equation of order two*, Proc. Amer. Math. Soc. **142**, 1193-1207 (2014),

these papers comprise a complete algorithm to compute the parameterized differential Galois group associated to a linear differential equation of the form

$$\frac{d^2}{dx^2} y + r_1 \frac{d}{dx} y + r_0 y = 0,$$

where r_1 and r_0 belong to $F(x)$, the field of rational functions in x with coefficients in a computable Π -field F of characteristic zero, and Π is a finite (possibly empty) set of pairwise commuting derivations.

In this note we illustrate how MAPLE can be used to carry out the computations required by this algorithm in some concrete examples with $F = \mathbb{C}(t)$ and $\Pi = \left\{ \frac{\partial}{\partial t} \right\}$.

The MAPLE computations carried out below may serve as the basis for a full implementation of the algorithm.

These examples follow the method proposed in [3], to first apply [2, 4] to compute the PPV group of an associated unimodular equation (i.e., an equation as above with $r_1 = 0$), and then use the data of the unimodular PPV group to recover the PPV group of the original equation.

The first three examples correspond to the first case of Kovacic's algorithm: the differential operator associated to the unimodular equation factors over $F^{alg}(x)$, which is equivalent to the statement that the PPV group is upper-triangular. This is the only case of Kovacic's algorithm in which the PPV group may be non-reductive (in all three examples considered here, the group is indeed non-reductive). The computation of the unipotent radical of the associated unimodular PPV group follows [2, Theorem 3.2] (see also [1]): if the reductive quotient is differentially constant, we apply [4, Section 2.1] to find the non-zero differential operator defining the unipotent radical as a subgroup of G_a , which operator arises as the solution to a hyperexponential creative telescoping problem; if, on the other hand, the reductive quotient is not differentially constant, then the unipotent radical is either trivial or all of G_a , and we apply [1, Lemma 4.3] to decide between these possibilities.

The last example concerns the second case of Kovacic's algorithm, where the differential operator corresponding to the associated unimodular equation only factors over a quadratic extension of $F^{alg}(x)$. In this case, the unimodular PPV group is of dihedral type (see [4, Section 2.2]), and the task of recovering the PPV group of the original equation is comparatively easier (see [3, Proposition 4.1]).

1. Example 1: Incomplete Gamma Function.

The PPV group in this example is computed via different methods in [1]. The incomplete Gamma function

$$\gamma(t, x) := \int_0^x s^{t-1} e^{-s} ds$$

satisfies the second-order linear differential equation

$$\frac{\partial^2}{\partial x^2} \gamma(t, x) - \frac{t-1-x}{x} \frac{\partial}{\partial x} \gamma(t, x) = 0.$$

So we set

```
> r[1] := -(t-1-x)/x; r[0]:=0;
```

$$r_1 := -\frac{t-1-x}{x}$$

$$r_0 := 0 \quad (1)$$

And then compute the coefficient q of the associated unimodular equation

$$\frac{d^2}{dx^2} y - qy = 0.$$

`> q := simplify(r[1]^2/4 + diff(r[1],x)/2 - r[0]);`

$$q := \frac{1}{4} \frac{t^2 - 2tx + x^2 + 2x - 1}{x^2} \quad (2)$$

Which is used as input for Kovacic's algorithm

`> with(DEtools): S := kovacicssols([-q,0,1],x);`

$$S := \left[\sqrt{x} x^{-\frac{1}{2}t} e^{\frac{1}{2}x}, \frac{1}{t(t+1)} \left(e^{\frac{1}{2}x} e^{-\frac{1}{2}x} \left(x^{\frac{1}{2}t-1} \text{WhittakerM}\left(\frac{1}{2}t+1, \frac{1}{2}t+\frac{1}{2}, x\right) t \right. \right. \right. \right. \\ \left. \left. \left. + x^{\frac{1}{2}t} \text{WhittakerM}\left(\frac{1}{2}t, \frac{1}{2}t+\frac{1}{2}, x\right) + x^{\frac{1}{2}t-1} \text{WhittakerM}\left(\frac{1}{2}t+1, \frac{1}{2}t+\frac{1}{2}, \right. \right. \right. \right. \\ \left. \left. \left. x\right) \right) \sqrt{x} x^{-\frac{1}{2}t} \right) \right] \quad (3)$$

We first compute the reductive quotient of the unimodular PPV group following [4, Lemma 3]. Since

`> type(S[1],radalgfun(ratpoly(complex,[x,t])));`
false (4)

this reductive quotient is infinite, and the operator defining it as a subgroup of G_m arises as the solution to the following creative telescoping problem:

`> u := simplify(diff(S[1],x)/S[1]); L[1] := Zeilberger(diff(u,t),t,x, Dt)[1];`

$$u := -\frac{1}{2} \frac{t-1-x}{x} \\ L_1 := Dt \quad (5)$$

Since

`> evalb(degree(L[1],Dt)>0);`
true (6)

the reductive quotient is not differentially constant, and it follows from [2, Theorem 3.2] (see also [1, Proposition 4.4]) that the computation of the unipotent radical of the Galois group is reduced to determining whether or not the following inhomogeneous equation admits a solution which is a rational expression in x .

`> UnipEq := diff(y(x),x) - 2*u*y(x) = 1;`

$$\text{UnipEq} := \frac{d}{dx} y(x) + \frac{(t-1-x)y(x)}{x} = 1 \quad (7)$$

`> evalb(nops(ratsols(UnipEq,y(x)))=1);`

true (8)

Since no rational solution exists, we set

```
> L[2]:=0;
```

$$L_2 := 0$$

(9)

which is equivalent to the statement that the unipotent radical is the whole additive group G_a [1, Lemma 4.3]. This concludes the computation of the unimodular PPV group. We now follow [3, Section 3] in recovering the original PPV group from these data. We first verify that

```
> z:= simplify(exp(int(-r[1]/2,x))); type(z,radalgfun(ratpoly
(complex,[x,t]));
```

$$z := e^{-\frac{1}{2}x} x^{\frac{1}{2}t - \frac{1}{2}}$$

false (10)

which implies that the quotient of the original PPV group by the determinant map is infinite. The operator defining this quotient as a subgroup of G_m arises as the solution to the following creative telescoping problem:

```
> L[3]:= Zeilberger(diff(r[1],t),t,x,Dt)[1];
```

$$L_3 := Dt$$

(11)

Since

```
> evalb(L[1]=L[3]);
```

true (12)

we check whether the following system admits a solution over the integers (see [3, Theorem 4.2(i) and Corollary 4.3(i)])

```
> Sys := {seq(E[a]*residue(u,x=p) + E[e]*residue(r[1]/2,x=p) - c[p]=
0, p in disjoint(u,x) union disjoint(r[1],x))};
```

$$Sys := \left\{ E_a \left(\frac{1}{2} - \frac{1}{2} t \right) + E_e \left(\frac{1}{2} - \frac{1}{2} t \right) - c_0 = 0 \right\}$$

(13)

In order to do this, we use the following procedure, which first finds rational solutions to the same system with $E_a = 1$ and then finds a solution over the integers with smallest possible gcd.

```
> R:=SolveTools[RationalCoefficients](add(C[p]*residue(u,x=p), p in
disjoint(u,x) union disjoint(r[1],x)), [-add(C[p]*residue(r[1],x=p)/2,
p in disjoint(u,x) union disjoint(r[1],x)), seq(C[p], p in disjoint(u,
x) union disjoint(r[1],x))]);
```

$$R := [-1, 0]$$

(14)

and set

```
> E[a]:= ilcm(op(map(denom,R))); E[e]:=R[1]*ilcm(op(map(denom,R)));
c_0:= R[2]*ilcm(op(map(denom,R)));
```

$$E_a := 1$$
$$E_e := -1$$
$$c_0 := 0$$

(15)

We then conclude that the parameterized differential Galois group associated to the incomplete Gamma function coincides with the set of matrices of the form (see [3, Theorem 4.2(1)]):

> **Matrix**([[**e(t)*a(t),e(t)*b(t)**],[**0,e(t)*a(t)^(-1)**]]);

$$\begin{bmatrix} e(t) a(t) & e(t) b(t) \\ 0 & \frac{e(t)}{a(t)} \end{bmatrix} \quad (16)$$

such that

> {**e(t)*a(t)<>0, eval(diffop2de(L[1],A(t),[Dt,t]),A(t)=diff(a(t),t)/a(t))=0, diffop2de(L[2],b(t),[Dt,t])=0, eval(diffop2de(L[3],B(t),[Dt,t]),B(t)=diff(e(t),t)/e(t))=0, a(t)^E[a]=e(t)^E[e]}**};

$$\left\{ \begin{aligned} 0 &= 0, \frac{\frac{d^2}{dt^2} a(t)}{a(t)} - \frac{\left(\frac{d}{dt} a(t)\right)^2}{a(t)^2} = 0, \frac{\frac{d^2}{dt^2} e(t)}{e(t)} - \frac{\left(\frac{d}{dt} e(t)\right)^2}{e(t)^2} = 0, a(t) = \frac{1}{e(t)}, \\ e(t) a(t) &\neq 0 \end{aligned} \right. \quad (17)$$

2. Example 2.

We now consider the equation

$$\frac{d^2}{dx^2} y - \frac{2t(tx-t+x)}{x(x-1)} \frac{d}{dx} y - \frac{t(2t^3x-t^3-2tx^2-2x^2+2x-1)}{x^2(x-1)^2} y = 0.$$

We begin by setting

> **restart: r[1] := -(2*t*(t*x-t+x))/(x*(x-1)); r[0] := -(t*(2*t^3*x-t^3-2*t*x^2-2*x^2+2*x-1))/(x^2*(x-1)^2);**

$$\begin{aligned} r_1 &:= -\frac{2t(tx-t+x)}{x(x-1)} \\ r_0 &:= -\frac{t(2t^3x-t^3-2tx^2-2x^2+2x-1)}{x^2(x-1)^2} \end{aligned} \quad (18)$$

And then compute the coefficient q of the associated unimodular equation

$$\frac{d^2}{dx^2} y - qy = 0.$$

> **q := simplify(r[1]^2/4 + diff(r[1],x)/2 - r[0]);**

$$q := \frac{t(t^3x^2+2t^2x^2-2t^2x-2tx-x^2+t+2x-1)}{x^2(x-1)^2} \quad (19)$$

This q is used as input for Kovacic's algorithm

```
> with(DEtools): S:= kovacicsols([-q,0,1],x);
   S := [(x-1)^{t^2} x^t, hypergeom([-2t+1, 2t^2], [2-2t], x) x^t (x-1)^{t^2} x^{-2t+1}]
```

(20)

We first compute the reductive quotient of the unimodular PPV group following [4, Lemma 3]. Since

```
> type(S[1],radalgfun(ratpoly(complex,[x,t])));
   false
```

(21)

this reductive quotient is infinite, and the operator defining it as a subgroup of G_m arises as the solution to the following creative telescoping problem:

```
> u := simplify(diff(S[1],x)/S[1]); L[1]:= Zeilberger(diff(u,t),t,x,
Dt)[1];
   u := \frac{t(tx+x-1)}{x(x-1)}
   L_1 := Dt^2
```

(22)

Since

```
> evalb(degree(L[1],Dt)>0);
   true
```

(23)

the reductive quotient is not differentially constant, and it follows from [2, Theorem 3.2] (see also [1, Proposition 4.4]) that the computation of the unipotent radical of the Galois group is reduced to determining whether or not the following inhomogeneous equation admits a solution which is a rational expression in x .

```
> UnipEq := diff(y(x),x) - 2*u*y(x) = 1;
   UnipEq := \frac{d}{dx} y(x) - \frac{2t(tx+x-1)y(x)}{x(x-1)} = 1
```

(24)

```
> evalb(nops(ratsols(UnipEq,y(x)))=1);
   true
```

(25)

Since no rational solution exists, we set

```
> L[2]:=0;
   L_2 := 0
```

(26)

which is equivalent to the statement that the unipotent radical is the whole additive group G_a [1, Lemma 4.3]. This concludes the computation of the unimodular PPV group. We now follow [3, Section 3] in recovering the original PPV group from these data. We first verify that

```
> z:= simplify(exp(int(-r[1]/2,x))); type(z,radalgfun(ratpoly
(complex,[x,t])));
   z := e^{(\ln(x)t + \ln(x-1))}
   false
```

(27)

which implies that the quotient of the original PPV group by the determinant map is infinite. The operator defining this quotient as a subgroup of G_m arises as the solution to the following creative telescoping problem:

```
> L[3] := Zeilberger(diff(r[1],t),t,x,Dt)[1];
                    L3 := Dt2 (28)
```

Since

```
> evalb(L[1]=L[3]);
                    true (29)
```

we check whether the following system admits a solution over the integers (see [3, Theorem 4.2(i) and Corollary 4.3(i)])

```
> Sys1 := {seq(E[a]*residue(u,x=p) + E[e]*residue(r[1]/2,x=p) - c[p]=
0, p in discont(u,x) union discont(r[1],x))};
                    Sys1 := { -t2 Ee + t Ea - c0 = 0, t2 Ea - t Ee - c1 = 0 } (30)
```

In order to do this, we use the following procedure, which first finds rational solutions to the same system with $E_a = 1$ and then finds a solution over the integers with smallest possible gcd.

```
> R:=SolveTools[RationalCoefficients](add(C[p]*residue(u,x=p), p in
discont(u,x) union discont(r[1],x)), [-add(C[p]*residue(r[1],x=p)/2,
p in discont(u,x) union discont(r[1],x)), seq(C[p], p in discont(u,
x) union discont(r[1],x))]);
                    R := FAIL (31)
```

Since no rational solution exists, there is no integer solution either. We must now decide whether the following linear system (from [3, Theorem 4.2(ii)])

```
> Sys2 := {seq(residue(sum(a[i-1]*diff(u,t$ i), i=1..degree(L[1],Dt))+
sum(b[j-1]*diff(r[1]/2,t$ j), j=1..degree(L[3],Dt)), x=p)=0, p in
discont(diff(u,t),x) union discont(diff(r[1],t),x))};
                    Sys2 := { 2 t a0 + 2 a1 - b0 = 0, -2 t b0 + a0 - 2 b1 = 0 } (32)
```

in the variables

```
> Var2 := {seq(a[i-1], i=1..degree(L[1],Dt)), seq(b[j-1], j=1..degree(L
[3],Dt))};
                    Var2 := { a0, a1, b0, b1 } (33)
```

admits a solution over F

```
> T:=table([op(solve(Sys2 union {a[1]=0,a[0]=1},Var2))]);
                    T := table( [ [ a1 = 0, a0 = 1, b0 = 2 t, b1 = -2 t2 +  $\frac{1}{2}$  ] ] ) (34)
```

Since it does admit a solution, we choose the unique solution with the smallest number of a_i non-zero and with the first non-zero a_i set to 1. We then conclude that the PPV group in this example coincides with the set of matrices of the form

```
> Matrix([ [e(t)*a(t), e(t)*b(t)], [0, e(t)*a(t)^(-1)] ]);
                    [ e(t) a(t)  e(t) b(t) ]
                    [ 0          e(t) ]
                    [          a(t) ] (35)
```

such that [3, Theorem 4.2(2)]

```
> {a(t)*e(t) <> 0, eval(diffop2de(L[1],A(t),[Dt,t]),A(t)=diff(a(t),t)
/a(t))=0, diffop2de(L[2],b(t),[Dt,t])=0, eval(diffop2de(L[3],B(t),
```

```
[Dt,t]),B(t)=diff(e(t),t)/e(t))=0, eval(sum(T[a[i]]*diff(A(t),t$),
i=0..(degree(L[1],Dt)-1)),A(t)=diff(a(t),t)/a(t))=eval(sum(T[b[j]]*
diff(B(t),t$),j=0..(degree(L[3],Dt)-1)),B(t)=diff(e(t),t)/e(t));
```

$$\left\{ \begin{aligned} 0 = 0, \frac{\frac{d}{dt} a(t)}{a(t)} &= \frac{2t \left(\frac{d}{dt} e(t) \right)}{e(t)} + \left(-2t^2 + \frac{1}{2} \right) \left(\frac{\frac{d^2}{dt^2} e(t)}{e(t)} - \frac{\left(\frac{d}{dt} e(t) \right)^2}{e(t)^2} \right), \\ \frac{\frac{d^3}{dt^3} a(t)}{a(t)} - \frac{3 \left(\frac{d^2}{dt^2} a(t) \right) \left(\frac{d}{dt} a(t) \right)}{a(t)^2} + \frac{2 \left(\frac{d}{dt} a(t) \right)^3}{a(t)^3} &= 0, \frac{\frac{d^3}{dt^3} e(t)}{e(t)} \\ - \frac{3 \left(\frac{d^2}{dt^2} e(t) \right) \left(\frac{d}{dt} e(t) \right)}{e(t)^2} + \frac{2 \left(\frac{d}{dt} e(t) \right)^3}{e(t)^3} &= 0, e(t) a(t) \neq 0 \end{aligned} \right\} \quad (36)$$

3. Example 3.

We now consider the equation

$$\frac{d^2}{dx^2} y - \frac{2t}{x} \frac{d}{dx} y + \frac{4t^2 + 4t + 1}{4x^2} y = 0.$$

We begin by setting

```
> restart: r[1] := -2*t/x; r[0] := (4*t^2 + 4*t + 1)/(4*x^2);
r1 := -2*t/x
r0 := 1/4 * (4*t^2 + 4*t + 1)/x^2
```

(37)

And then compute the coefficient q of the associated unimodular equation

$$\frac{d^2}{dx^2} y - qy = 0.$$

```
> q := simplify(r[1]^2/4 + diff(r[1],x)/2 - r[0]);
q := -1/(4*x^2)
```

(38)

This q is used as input for Kovacic's algorithm

```
> with(DEtools): S := kovacicsols([-q,0,1],x);
S := [sqrt(x), sqrt(x) ln(x)]
```

(39)

We first compute the reductive quotient of the unimodular PPV group following [4, Lemma 3]. Since

```
> type(S[1],radalgfun(ratpoly(complex,[x,t])));
true
```

(40)

we conclude that the reductive quotient of the parameterized differential Galois group for the associated unimodular equation is finite of order

$$\begin{aligned} > \mathbf{E[a]} := \mathbf{Algebraic[Degree]({convert(S[1],RootOf})}); \\ & \quad E_a := 2 \end{aligned} \tag{41}$$

It will be convenient to define

$$\begin{aligned} > \mathbf{u} := \mathbf{simplify(diff(S[1],x)/S[1]); L[1]:=1; \\ & \quad u := \frac{1}{2x} \\ & \quad L_1 := 1 \end{aligned} \tag{42}$$

In order to decide whether or not the unipotent radical is trivial, we must decide whether or not the following inhomogeneous equation admits a solution which is a rational expression in x

$$\begin{aligned} > \mathbf{UnipEq} := \mathbf{diff(y(x),x) - 2*u*y(x) = 1; \\ & \quad UnipEq := \frac{d}{dx} y(x) - \frac{y(x)}{x} = 1 \end{aligned} \tag{43}$$

$$\begin{aligned} > \mathbf{evalb(nops(ratsols(UnipEq,y(x)))=1); \\ & \quad true \end{aligned} \tag{44}$$

Since there is no such solution, we know that the unipotent radical is not trivial [1, Lemma 4.3]. Since

$$\begin{aligned} > \mathbf{evalb(diff(u,t)=0); \\ & \quad true \end{aligned} \tag{45}$$

the following telescoping problem has a solution, and its output is the differential operator that defines the unipotent radical as a subgroup of G_a [4, Section 2.1, p. 1200].

$$\begin{aligned} > \mathbf{L[2]} := \mathbf{Zeilberger(1/S[1]^2,t,x,Dt)[1]; \\ & \quad L_2 := Dt \end{aligned} \tag{46}$$

This concludes the computation of the unimodular PPV group. We now follow [3, Section 3] in recovering the original PPV group from these data. We first verify that

$$\begin{aligned} > \mathbf{z} := \mathbf{simplify(exp(int(-r[1]/2,x))}; \mathbf{type(z,radalgfun(ratpoly} \\ & \quad \mathbf{(complex,[x,t]))}); \\ & \quad z := x^t \\ & \quad false \end{aligned} \tag{47}$$

which implies that the quotient of the original PPV group by the determinant map is infinite. The operator defining this quotient as a subgroup of G_m arises as the solution to the following creative telescoping problem:

$$\begin{aligned} > \mathbf{L[3]} := \mathbf{Zeilberger(diff(r[1],t),t,x,Dt)[1]; \\ & \quad L_3 := Dt \end{aligned} \tag{48}$$

By [3, Corollary 4.3(i)], it is not necessary to check whether [3, Theorem 4.2(i)] holds. Similarly, it follows from (45) that [3, Theorem 4.2(ii)] does not hold. Since

$$\begin{aligned} > \mathbf{evalb(E[a]<=2); \\ & \quad true \end{aligned} \tag{49}$$

we need to decide whether or not the linear system from [3, Theorem 4.2(iii)]

$$\begin{aligned} &> \text{Sys} := \{ \text{seq}(\text{sum}(a[i] * \text{diff}(\text{residue}(1/S[1]^2, x=p), t\$i), i=0..(\text{degree}(L[2], Dt)-1)) + \text{sum}(b[j] * \text{diff}(\text{residue}(\text{diff}(r[1], t), x=p), t\$j), j=0..(\text{degree}(L[3], Dt)-1))/2=0, p \text{ in } \text{discont}(1/S[1]^2, x) \text{ union } \text{discont}(\text{diff}(r[1], t), x)) \}; \\ & \text{Sys} := \{a_0 - b_0 = 0\} \end{aligned} \quad (50)$$

in the variables

$$\begin{aligned} &> \text{Var} := \{ \text{seq}(a[i], i=0..(\text{degree}(L[2], Dt)-1)), \text{seq}(b[j], j=0..(\text{degree}(L[3], Dt)-1)) \}; \\ & \text{Var} := \{a_0, b_0\} \end{aligned} \quad (51)$$

admits a solution in F :

$$\begin{aligned} &> \text{T} := \text{table}([\text{op}(\text{solve}(\text{Sys union } \{a[\text{degree}(L[2], Dt)-1]=1\}, \text{Var}))]); \\ & \text{T} := \text{table}([b_0 = 1, a_0 = 1]) \end{aligned} \quad (52)$$

We then conclude that the PPV group in this example coincides with the set of matrices of the form

$$\begin{aligned} &> \text{Matrix}([\text{e}(t) * a(t), \text{e}(t) * b(t)], [0, \text{e}(t) * a(t)^{-1}]); \\ & \begin{bmatrix} e(t) a(t) & e(t) b(t) \\ 0 & \frac{e(t)}{a(t)} \end{bmatrix} \end{aligned} \quad (53)$$

such that [3, Theorem 4.2(3)]

$$\begin{aligned} &> \{ \text{e}(t) * a(t) \neq 0, \text{eval}(\text{diffop2de}(L[1], A(t), [Dt, t]), A(t) = \text{diff}(a(t), t) / a(t)) = 0, \text{diffop2de}(L[2], b(t), [Dt, t]) = 0, \text{eval}(\text{diffop2de}(L[3], B(t), [Dt, t]), B(t) = \text{diff}(e(t), t) / e(t)) = 0, a(t)^E[a]=1, \text{sum}(T[a[i]] * \text{diff}(b(t), t\$i), i=0..(\text{degree}(L[2], Dt)-1)) = \text{eval}(\text{sum}(T[b[j]] * \text{diff}(B(t), t\$j), j=0..(\text{degree}(L[3], Dt)-1)), B(t) = \text{diff}(e(t), t) / e(t)) \}; \\ & \left\{ \begin{aligned} a(t)^2 = 1, \frac{d}{dt} a(t) = 0, \frac{d^2}{dt^2} e(t) - \frac{\left(\frac{d}{dt} e(t)\right)^2}{e(t)^2} = 0, b(t) = \frac{d}{dt} e(t), \frac{d}{dt} b(t) = 0, \\ e(t) a(t) \neq 0 \end{aligned} \right. \end{aligned} \quad (54)$$

4. Example 4.

We now consider the equation

$$\frac{d^2}{dx^2} y + \frac{2t}{x-1} \frac{d}{dx} y + \frac{16t^2 x^2 - 16t^2 x - 16tx^2 + 3x^2 - 2x + 3}{16x^2(x-1)^2} y = 0.$$

We begin by setting

$$\begin{aligned} &> \text{restart: } r[1] := 2*t/(x-1); r[0] := (16*t^2*x^2 - 16*t^2*x - 16*t*x^2 + 3*x^2 - 2*x + 3)/(16*x^2*(x-1)^2); \end{aligned}$$

$$r_1 := \frac{2t}{x-1}$$

$$r_0 := \frac{1}{16} \frac{16t^2x^2 - 16t^2x - 16tx^2 + 3x^2 - 2x + 3}{x^2(x-1)^2} \quad (55)$$

And then compute the coefficient q of the associated unimodular equation

$$\frac{d^2}{dx^2} y - qy = 0.$$

> q := simplify(r[1]^2/4 + diff(r[1],x)/2 - r[0]);

$$q := \frac{1}{16} \frac{16t^2x - 3x^2 + 2x - 3}{x^2(x-1)^2} \quad (56)$$

This q is used as input for Kovacic's algorithm

> with(DEtools): S:= kovacicols([-q,0,1],x);

$$S := \left[\sqrt{x-1} x^{1/4} \left(\frac{-1+\sqrt{x}}{\sqrt{x}+1} \right)^t, \sqrt{x-1} x^{1/4} \left(\frac{\sqrt{x}+1}{-1+\sqrt{x}} \right)^t \right] \quad (57)$$

We now define

> u:=simplify(diff(S[1],x)/S[1]);

$$u := \frac{1}{4} \frac{4xt + 3x^{3/2} - \sqrt{x}}{x^{3/2}(x-1)} \quad (58)$$

Since the degree of the minimal polynomial

> p:=gfun[algfuntoalgeq](u,z(x));

$$p := (-16x^4 + 32x^3 - 16x^2)z^2 + (24x^3 - 32x^2 + 8x)z + 16t^2x - 9x^2 + 6x - 1 \quad (59)$$

of u over $\mathbb{C}(x, t)$ is

> degree(p,z);

$$2 \quad (60)$$

we are in case II of Kovacic's algorithm (see [3, p. 45, col. 1] and [4, Section 2.2]). To conclude the computation of the PPV group of the associated unimodular equation, we follow [4, Section 2.2]

> L[1]:= Zeilberger(diff(u,t),t,x,Dt)[1];

$$L_1 := Dt \quad (61)$$

Now we follow [3, Section 4.1] in computing the PPV group of the original equation.

First, we verify that

> z:=simplify(exp(int(-r[1]/2,x))); type(z,radalgfun(ratpoly(complex,[x,t])));

$$z := (x-1)^{-t}$$

false

$$(62)$$

which implies that the quotient of the original PPV group by the determinant map is infinite. The operator defining this quotient as a subgroup of

G_m arises as the solution to

the following creative telescoping problem:

$$\begin{aligned} > L[2] := \text{Zeilberger}(\text{diff}(r[1], t), t, x, Dt)[1]; \\ & \qquad \qquad \qquad L_2 := Dt \end{aligned} \tag{63}$$

We now conclude from [3, Proposition 4.1] that the PPV group of the original equation coincides with the set of matrices in

$$\begin{aligned} > \{ \text{Matrix}([[e(t)*a(t), 0], [0, e(t)*a(t)^{-1}]]), \text{Matrix}([[0, -e(t)*a(t) \\], [e(t)*a(t)^{-1}, 0]]) \}; \\ & \left\{ \left[\begin{array}{cc} 0 & -e(t) a(t) \\ \frac{e(t)}{a(t)} & 0 \end{array} \right], \left[\begin{array}{cc} e(t) a(t) & 0 \\ 0 & \frac{e(t)}{a(t)} \end{array} \right] \right\} \end{aligned} \tag{64}$$

such that

$$\begin{aligned} > \{ a(t)*e(t) \neq 0, \text{eval}(\text{diffop2de}(L[1], A(t), [Dt, t]), A(t) = \text{diff}(a(t), t)/a \\ (t)) = 0, \text{eval}(\text{diffop2de}(L[2], B(t), [Dt, t]), B(t) = \text{diff}(e(t), t)/e(t)) = 0 \} \\ ; \\ & \left\{ \frac{\frac{d^2}{dt^2} a(t)}{a(t)} - \frac{\left(\frac{d}{dt} a(t)\right)^2}{a(t)^2} = 0, \frac{\frac{d^2}{dt^2} e(t)}{e(t)} - \frac{\left(\frac{d}{dt} e(t)\right)^2}{e(t)^2} = 0, e(t) a(t) \neq 0 \right\} \end{aligned} \tag{65}$$

>